

**USING SAS/GRAPH AT NIH,
VERSION 6**

**Statistical Support Staff
Office of Computing Resources and Services
Division of Computer Research and Technology
National Institutes of Health**

May 13, 1997

TABLE OF CONTENTS

1. INTRODUCTION	1
1.1 NIH GRAPHICS DEVICES SUPPORTED IN SAS/GRAPH	1
1.2 MODES OF USING SAS/GRAPH	2
2. USING SAS/GRAPH IN BATCH MODE	2
2.1 USING THE LASER PRINTERS	2
2.1.1 PAPER	3
2.1.2 GRAY SHADES AVAILABLE	3
2.1.3 WRITING AND RUNNING THE PROGRAM	4
2.1.4 GETTING A HARDCOPY OF THE OUTPUT	5
2.1.5 EXAMPLES	6
2.2 USING THE POSTSCRIPT DRIVER	9
2.2.1 PAPER	9
2.2.2 GRAY SHADES AVAILABLE	9
2.2.3 CREATING AND PRINTING THE POSTSCRIPT FILE	10
2.2.4 EXAMPLE	11
3. USING SAS/GRAPH IN INTERACTIVE MODE	12
3.1 INTERACTIVE LINE MODE	12
3.2 LINE MODE SAMPLE SESSION	14
3.3 INTERACTIVE FULL-SCREEN MODE	16
3.4 FULL-SCREEN SAMPLE SESSION	17
4. SPECIAL SAS/GRAPH DATA SETS	18
4.1 MAP DATA SETS	18
4.2 TEMPLATE CATALOG	18
4.3 SAMPLE LIBRARY	19
4.4 ANNOTATE MACROS	19
5. TRANSFERRING SAS/GRAPH FILES TO THE PC	20

1. Introduction

SAS/GRAPH is the graphics software product of the SAS System. It can be used to display data in various forms. It can produce plots, bar charts, pie charts, maps, text charts and other graphics output. It offers the user many options with which to enhance their text and design. Its procedures can be used together with procedures of other SAS products.

This document explains how to use version 6 of SAS/GRAPH with graphics devices supported by the IBM mainframe. It is intended to be used as a supplement to the “*SAS/GRAPH Software*” manuals available through WYLBUR’s ENTER PUBWARE command or from the Technical Information Office (301-594-DCRT).

In the examples given in this document the user's initials are represented by “*iii*”, the account by “*aaaa*”, the box number by “*999*” and the last name by “*lastname*”. For assistance with SAS/GRAPH you may call 301-594-DCRT.

1.1 NIH Graphics Devices Supported in SAS/GRAPH

At NIH, the following graphics devices are available on the IBM mainframe and can be used to produce SAS/GRAPH output with version 6 of SAS:

- IBM 3900 laser printers,
- IBM 3827 laser printers, and,
- a PostScript printing facility.

To obtain a complete list of the graphics devices that SAS supports run a batch program with the following statements:

```
PROC GDEVICE CATALOG=SASHELP.DEVICES NOFS;  
LIST _ALL_;
```

If you use SAS in full-screen mode, submit the following statements instead:

```
PROC GDEVICE CATALOG=SASHELP.DEVICES;  
LIST _ALL_;  
RUN;
```

1.2 Modes of Using SAS/GRAPH

SAS/GRAPH procedures can be used in two modes:

- in batch programs using WYLBUR, or,
- in interactive mode using TSO.

The laser printers are used in batch mode. If your workstation can emulate a graphics terminal you may use interactive mode to view the results on the screen.

2. Using SAS/GRAPH in Batch Mode

To send SAS/GRAPH output to the laser printers or the PostScript printer you use SAS in batch mode by connecting to WYLBUR. This means you first write a SAS program that includes some JCL (Job Control Language) statements and then use WYLBUR's RUN command to run it.

More information on using SAS in batch mode can be found in the document "*Using SAS at NIH: Batch Mode*" available through WYLBUR's ENTER PUBWARE command or from the Technical Information Office.

2.1 Using the Laser Printers

The IBM 3900 and IBM 3827 laser printers produce high quality output. Both support fifteen different gray shades, including black and white, though the colors are sharper on the 3900. They use different types of paper.

The cost of using SAS/GRAPH to produce graphs on the 3900 and the 3827 printers can be high depending on the complexity of the graphs. Printing the graphics output is normally inexpensive. What may be high is the cost of running the program. Consider running your programs overnight to receive a 60 percent rate reduction (use a /*DISCOUNT statement after the JOB statement).

The output produced from the laser printers is normally placed in your box at building 12A in about three hours.

2.1.1 Paper

Graphics output routed to the 3900 is printed on laser-standard paper which is 11 by 8 1/2 inches. This paper has 1/2 inch easy-strip margins. By default, output is produced in landscape format (horizontal).

On the 3827, graphics output is produced on single sheets of paper of 11 by 8 1/2 inches. By default, output is produced in portrait format (vertical).

To rotate the graphs use the ROTATE option in a GOPTIONS statement as in example 3 of section 2.1.5.

2.1.2 Gray Shades Available

The 3900 and the 3827 support the same colors: black and various shades of gray. What follows is a list of the names of the colors as they should be requested in SAS programs. These names are used to represent the different gray shades. They are listed from darkest to lightest.

Color Names
black
brown
red
dab
pink
dag
green
orange
yellow
purple
gold
cyan
libg
gray
white

For an illustration of these gray shades see “*Gray-Scale Shading With SAS/GRAPH on the Laser Printers*” in Interface 151.

2.1.3 Writing and Running the Program

To route output to the IBM 3900 or 3827, write a program that starts with the following JCL statements:

```
//iii JOB (aaaa,999,class),lastname  
// JCLLIB ORDER=ZABCRUN.PROCLIB  
// EXEC SAS  
//SYSIN DD *
```

The first line is the JOB statement. It should specify your initials, account, box number, job class and your last name. An extra statement is needed to use map data sets (see example 2 in section 2.1.5).

After the JCL statements, write the SAS statements. Before any graphics procedure, include a GOPTIONS statement to tell SAS where to send the graphics output.

To route output to the IBM 3900 specify:

```
GOPTIONS DEVICE=GDDMFAM4 GDDMN=IBM3900 ;
```

To route output to the IBM 3827 specify:

```
GOPTIONS DEVICE=GDDMFAM4 GDDMN=IBM3827 ;
```

Examples 1 and 2 of section 2.1.5 demonstrate the use of these statements.

To run the SAS program use the UNN option in WYLBUR's RUN command. For example, you may use the command:

```
RUN UNN HOLD NTF
```

It is a good idea to use the HOLD and NTF (notify) options but they are not necessary. The HOLD option specifies that the output should not be printed immediately after the job is done, but placed in "output hold". The NTF option specifies that you should be notified when the job completes execution.

If the HOLD option is specified, the FETCH command may be used after the job finishes running. It puts the non-graphics output into your active file. Just enter the word FETCH followed by the job number assigned to the job when the RUN command was entered.

The graphics output cannot be viewed at the terminal. If the terminal you are using is a graphics terminal and you want to display the graphs on the screen use SAS in interactive mode.

After FETCHing the job you may list it with WYLBUR's LIST command. It is always a good idea to FETCH the output and list it at the terminal before printing to check for errors.

2.1.4 Getting a Hardcopy of the Output

If the HOLD option is not specified in the RUN command, the graphics and non-graphics output are sent automatically to the printers when the job finishes execution and you will not be able to list the SAS output on the screen to check for errors.

If the HOLD option is used you may obtain a hardcopy of the graphics and non-graphics output using WYLBUR's PRINT command. Enter the word PRINT followed by the job number. The output will be placed in your box in about three hours. You will receive the graphics output and the non-graphics output separately.

To get multiple copies of the graphics and non-graphics output, the COPIES= option of the RUN command can be used or a /*JOBOUT statement with the COPIES= option. For example, if you want two copies, enter the following command after your JOB statement:

```
/ *JOBOUT COPIES=2
```

2.1.5 Examples

Example 1: Sample WYLBUR session and sending output to the 3900 laser printer.

```
? COLLECT
```

```
1.  ? //iii JOB (aaaa,999,A),lastname
2.  ? // JCLLIB ORDER=ZABCRUN.PROCLIB
3.  ? // EXEC SAS
4.  ? //SYSIN DD *
5.  ?
6.  ? GOPTIONS DEVICE=GDDMFAM4 GDDMN=IBM3900;
7.  ?
8.  ? DATA MIGRA;
9.  ? LENGTH COUNTRY $ 13;
10. ? INPUT COUNTRY $ & PEOPLE @@;
11. ? CARDS;
12. ? Europe 62.5 Asia 268.2 North America 207.7
13. ? South America 41.9 Other 21.4
14. ? ;
15. ? PROC GCHART;
16. ? TITLE F=CENTX 'IMMIGRANTS BY COUNTRY OF BIRTH: 1986';
17. ? PATTERN1 V=PSOLID C=GOLD;
18. ? PATTERN2 V=PSOLID C=ORANGE;
19. ? PATTERN3 V=PSOLID C=GRAY;
20. ? PATTERN4 V=PSOLID C=RED;
21. ? PATTERN5 V=PEMPTY C=BLACK;
22. ? PIE COUNTRY / MIDPOINTS='Asia' 'North America' 'Europe'
23. ? 'South America' 'Other' SLICE=ARROW
24. ? VALUE=ARROW PERCENT=ARROW SUMVAR=PEOPLE
25. ? NOHEADING CTEXT=BLACK COUTLINE=BLACK;
26. ? *** <== Press the BREAK key.

? SAVE AS MIGRA.SAS
"MIGRA.SAS" SAVED AND CATALOGED ON FILE
? RUN UNN HOLD NTF
JOB 100 III SUBMITTED
? ***
JOB 100 III ENDED AT NIHCU
? FETCH 100 CLR
? LIST
```

The non-graphics output will be listed at the terminal. If there are any error messages, go back to the program and modify it. Then run it again. Otherwise, you may enter the PRINT command to print the job.

```
? PRINT 100
JOB 100 III OUTPUT ROUTING IS NOW GO
```

Example 2: Sending output to the 3827 laser printer.

```
//iii JOB (aaaa,999,A),lastname
// JCLLIB ORDER=ZABCRUN.PROCLIB
// EXEC SAS
//SYSIN DD *

GOPTIONS DEV=GDDMFAM4 GDDMN=IBM3827;

LIBNAME MAP 'ZABCRUN.SAS609.MAPS' DISP=SHR;

DATA SPEED65;
LENGTH STATE 2;
INPUT STCODE $ @@;
MPH65='N';
STATE=STFIPS(STCODE);
CARDS;
AK CT DE DC HI MD MA NJ NY PA RI
;

PROC GMAP MAP=MAP.US ALL;
TITLE 'NO 65 MILES PER HOUR';
PATTERN1 V=M2X C=RED;
PATTERN2 V=E C=BLACK;
ID STATE;
CHORO MPH65 / NOLEGEND;
```

Example 3: Rotating the graphics output 90 degrees.

```
//iii JOB (aaaa,999,A),lastname
// JCLLIB ORDER=ZABCRUN.PROCLIB
// EXEC SAS
//SYSIN DD *

GOPTIONS DEV=GDDMFAM4 GDDMN=IBM3827 ROTATE;

DATA FOOD;
INPUT YEAR CONSUME @@;
LABEL YEAR='Year';
CARDS;
80 86.7 81 83.9 82 83.9 83 88.2 84 87.5 85 86.5 86 91.7
;
TITLE F=ZAPF H=2 'Fresh Fruit Consumption: 1980 to 1986';

PROC GPLOT;
SYMBOL1 C=BLACK W=5 I=JOIN V=NONE;
AXIS1 LABEL=(F=ZAPF H=1.5 A=90 R=0 'Consumption Per Capita')
      VALUE=(F=ZAPF H=1.5);
AXIS2 LABEL=(F=ZAPF H=1.5) VALUE=(F=ZAPF H=1.5);
PLOT CONSUME*YEAR / VAXIS=AXIS1 HAXIS=AXIS2 HMINOR=0 FRAME;

PROC GCHART;
VBAR YEAR / DISCRETE SUMVAR=CONSUME RAXIS=AXIS1;
AXIS1 LABEL=(A=90 R=0 'Consumption Per Capita');
```

2.2 Using the PostScript Driver

Version 6 of SAS/GRAPH includes a PostScript driver with which one can create a PostScript file that can be used as input to the IBM POSTSCRIP utility to print the file on the 3827 laser printers. Alternatively, the file may be downloaded to another computer to be viewed, printed or copied to a document. The file created by SAS is a binary file.

2.2.1 Paper

Output is printed to single sheets of paper of 8 1/2 by 11 inches. By default, output is produced in portrait format. To rotate the graph use the ROTATE option in a GOPTIONS statement as in example 3 of section 2.1.6.

2.2.2 Gray Shades Available

The colors supported for the PostScript driver are black and eight gray shades on a white background. The gray shades are coded with the names below. They are listed from darkest to lightest.

Color Names
black
gray22
gray33
gray44
gray66
gray88
grayaa
graycc

2.2.3 Creating and Printing the PostScript file

To create a PostScript file using SAS/GRAPH include the statements below before any graphics procedure and after the JCL (see section 2.1.3).

```
FILENAME GSASFILE 'aaaaiii.psname' UNIT=FILE
  DISP=(MOD,CATLG)SPACE=(TRK,(50,1),RLSE) RECFM=VB
  LRECL=80 BLKSIZE=8000;

GOPTIONS DEVICE=PS GACCESS=GSASFILE GPROTOCOL=SASGPASC
  GSFLLEN=80 NOPROMPT HANDSHAKE=NONE TRANTAB=GTABCMS;
```

You may change the SPACE parameter.

At the end of your program add the following two lines:

```
// EXEC POSTSCRP,NAME='aaaaiii.psname'
// EXEC DSSCR,NAME='aaaaiii.psname'
```

Here, “*psname*” is any valid WYLBUR name. This is where the PostScript file will be saved.

To run the program and produce a hardcopy of the graph see the explanation in section 2.1.3 and 2.1.4.

2.2.4 Example

The following SAS program creates a PostScript file called TRAFFIC.PS under account and initials “aaaaiii”. The file is used as input to the POSTSCRIP utility. At the end of the program, we delete the file since it's no longer needed.

```
//iii JOB (aaaa,999,A),lastname
// JCLLIB ORDER=ZABCRUN.PROCLIB
// EXEC SAS
//SYSIN DD *

FILENAME GSASFILE 'aaaaiii.TRAFFIC.PS' UNIT=FILE
        DISP=(MOD,CATLG) SPACE=(TRK,(20,1),RLSE) RECFM=VB LRECL=80
        BLKSIZE=8000;

GOPTIONS DEV=PS GACCESS=GSASFILE GPROTOCOL=SASGPASC
        GSFLLEN=80 NOPROMPT HANDSHAKE=NONE TRANTAB=GTABCMS;

DATA TEST;
INPUT YEAR TRAFFIC @@;
CARDS;
90 120 91 140 93 180
;

PATTERN1 V=S C=GRAY22;
PATTERN2 V=S C=GRAY66;
PATTERN3 V=S C=GRAYCC;
PROC GCHART;
VBAR YEAR / DISCRETE SUMVAR=TRAFFIC PATTERNID=MIDPOINT;

SYMBOL I=JOIN F=SIMPLEX C=BLACK;
PROC GPLOT;
PLOT TRAFFIC*YEAR;
// EXEC POSTSCRIP,NAME='aaaaiii.TRAFFIC.PS'
// EXEC DSSCR,NAME='aaaaiii.TRAFFIC.PS'
```

3. Using SAS/GRAPH in Interactive Mode

If you use a software package that emulates a graphics terminal you may use SAS/GRAPH in interactive mode to view the graphs on the screen. Examples of these are VersaTerm-PRO (line mode) and tn3270 (full-screen mode) on the Macintosh, and Kermit (line mode) on the personal computer.

To use SAS in interactive mode you log on to TSO (not WYLBUR). TSO offers two modes of execution:

- line mode, and,
- full-screen mode.

In line mode you enter SAS statements sequentially in response to prompts from SAS.

In full-screen mode you enter the SAS Display Manager System where you write your program in a program editor window.

To learn more about running SAS in interactive mode see the handout “*Using SAS at NIH: Interactive Mode*”, the “*SAS Companion to the MVS Environment*” and the “*SAS Language*” guides. These are available at the Technical Information Office or through WYLBUR's ENTER PUBWARE.

3.1 Interactive Line Mode

To invoke SAS in interactive line mode connect to TSO. Check your emulation software documentation for help on connecting to TSO. The dialup access telephone number for line mode TSO is 301-402-2223, and its Internet hostname is tso.cu.nih.gov.

At TSO's READY prompt enter the following command to start a SAS session:

```
EX 'ZABCRUN.SAS.CLIST(SAS)'
```

Then you enter SAS statements sequentially in response to prompts from SAS.

In your program, include a GOPTIONS statement as follows:

```
GOPTIONS TRANTAB=GTAB0043 DEVICE=devname;
```

where “*devname*” is the name of the SAS graphics driver you are requesting (e.g. TEK4014). The TRANTAB option is required.

Instead of entering the statements in the SAS session you may collect them first using WYLBUR and then bring them into the SAS session. This makes editing much easier. You may follow these steps:

1. In WYLBUR, write the SAS program and save it in LRECL form. Include a RUN statement at the end of the last PROC. If the program reads data from an external file, save that file in LRECL form also. (The LRECL should be at least equal to the length of the longest line in the file.)
2. Logon to TSO.
3. After the READY prompt invoke SAS.
4. To run the program, enter the statement:

```
%INCLUDE 'aaaaiii.pgm' ;
```

where “aaaaiii.pgm” is the name of the SAS program preceded by your account and initials. (%INCLUDE may be abbreviated to %INC.)

5. To end the SAS session enter the statement:

```
ENDSAS ;
```

If the program has errors you may enter the statement:

```
X WYL ;
```

to go from the SAS session to WYLBUR, where you can correct the program. After changing and resaving the program, end the WYLBUR session to go back to SAS. Then use the %INCLUDE to bring in the program.

3.2 Line Mode Sample Session

Connect to WYLBUR and write the SAS program:

```
? collect

1.  ? DATA LOTTERY;
2.  ? INFILE 'aaaaiii.LOTTERY.DAT';
3.  ? INPUT WHAT $ PERCENT;
4.  ? RUN;
5.  ?
6.  ? GOPTIONS TRANTAB=GTAB0043 DEV=TEK4105;
7.  ?
8.  ? PROC GCHART;
9.  ? TITLE 'What we would do if we won $1,000,000';
10. ? HBAR WHAT / SUMVAR=PERCENT NOSTATS RAXIS=AXIS1
11. ?             MAXIS=AXIS2;
12. ? AXIS1 LABEL=('Percent of People');
13. ? AXIS2 VALUE=(T=5 'Start own' J=L 'business'
14. ?             T=4 'Quit working'
15. ?             T=3 'Go back to' J=L 'school'
16. ?             T=2 'Try a new' J=L 'profession'
17. ?             T=1 'Other') LABEL=NONE;
18. ? RUN;      <=== Include the RUN statement.
19. ? ***      <=== Press the BREAK key.

? save as lottery.sas lrecl=50
? end clr
```

After you logon to TSO allocate the program and external data set (if any) and invoke SAS:

```
READY
ex 'zabcrun.sas.clist(sas)'
```

Some general information will be displayed at this point. Then you get a 1?

```
1? %inc 'aaaaiii.LOTTERY.SAS';
NOTE: The data set WORK.LOTTERY has 5 observations and 2
variables.
NOTE: The DATA statements used 0.03 CPU seconds and 3632K.
Please press return after each bell to continue.
```

At this point, the graph will be displayed on the screen or routed to the plotter depending on the device being used. Press return to clear the screen.

```
23? endsas;
```

NOTE: SAS Institute, SAS Campus Drive, Cary, NC USA 27512-
2414
READY

3.3 Interactive Full-Screen Mode

To invoke SAS in full-screen mode connect to TSO. Check your emulation software documentation for help on connecting to TSO. The dialup access telephone number for full-screen TSO is 301-402-2227, and it's Internet hostname is tn3270.cu.nih.gov.

If you are using tn3270 for the Mac make sure the hostname in the **Settings** menu is set to:

```
tn3270.cu.nih.gov:IBM-3278-2-E
```

At TSO's READY prompt enter the following command to start a SAS session:

```
EX 'ZABCRUN.SAS.CLIST(SAS)'
```

At this point the SAS Display Manager System appears with a PROGRAM EDITOR window and a LOG window. Write your program in the PROGRAM EDITOR window.

In your program, include a GOPTIONS statement as follows:

```
GOPTIONS DEVICE=devname;
```

where "*devname*" is the name of the SAS graphics driver you are requesting (see example). If you are using tn3270 for the Mac you may use the GDDMPCGX driver.

Include a RUN statement after the last PROC or DATA step.

When you are finished, enter the command SUBMIT on the program editor's command line (i.e. at the arrow) to run the program. The graph is then displayed on the screen. Press return to go back to the program editor.

Enter the BYE command on any command line to exit SAS.

3.4 Full-Screen Sample Session

Here we run a PROC GSLIDE using the Display Manager System and use the GDDMPCGX device driver.

```
+LOG-----+
| Command ==>                                     |
|                                                  |
|                                                  |
|                                                  |
|-----+
+PROGRAM EDITOR-----+
| Command ==>  SUBMIT                               |
|                                                  |
| 00001 GOPTIONS DEVICE=GDDMPCGX;                 |
| 00002                                             |
| 00003 PROC GSLIDE;                               |
| 00004 TITLE F=SWISSL 'USING SAS/GRAPH AT NIH';   |
| 00005 RUN;                                       |
| 00006                                           |
+-----+
```

4. Special SAS/GRAPH Data Sets

Several data sets are available at NIH for different applications using SAS/GRAPH. They are described in the subsequent sections.

4.1 Map Data Sets

SAS/GRAPH provides an extensive library of SAS data sets to draw maps of the US and most other countries. To access them in batch mode, include the following LIBNAME statement in your program:

```
LIBNAME MAP 'ZABCRUN.SAS609.MAPS' DISP=SHR;
```

In interactive mode this library gets allocated automatically with a libref of MAP.

To get a complete list of the maps available run a SAS program with the following statements:

```
LIBNAME MAP 'ZABCRUN.SAS609.MAPS' DISP=SHR;  
PROC CONTENTS DATA=MAP._ALL_ NODS;
```

In some of these SAS data sets the coordinates are unprojected, i.e. they are stored as latitude and longitude of a sphere. When they are plotted on a two-dimensional plane, the resulting map will be distorted. PROC GPROJECT can be used first to create a map data set where the coordinates are projected onto a two-dimensional plane.

Example 2 of section 2.1.5 illustrates the use of a map data set.

4.2 Template Catalog

Twenty one templates are available for use with PROC GREPLAY. The template catalog is automatically available in batch and interactive mode. To access it, enter the statement:

```
TC SASHELP.TEMPLT;
```

after the PROC GREPLAY statement. To list the contents of the catalog follow the TC statement with:

```
LIST TC;
```

4.3 Sample Library

Many of the examples given in the SAS/GRAPH guides are saved in a partitioned data set (PDS). Each member of the PDS is a SAS program. The PDS name is

```
ZABCRUN.SAS609.SAMPLE
```

To view the program that created a figure in a SAS/GRAPH guide enter the following WYLBUR command then list it:

```
USE FR &ZABCRUN.SAS609.SAMPLE(member) ON CAT
```

where “*member*” is the code that appears at the bottom right hand corner of the figure in the “SAS/GRAPH Software” manuals.

4.4 Annotate Macros

To use the annotate macros discussed in Chapter 19 of the “SAS/GRAPH Software, Volume 1” guide, include the following statement in your program:

```
%ANNOMAC ;
```

After this statement you may call any of the macros by entering a percent sign, then the macro name followed by a semicolon. They can be called only from a DATA step.

5. Transferring SAS/GRAPH Files to the PC

You can produce graphs on the mainframe, store them in a graphics stream file (GSF), download them to your PC, then send them to the local PC printer or include them in a document. Graphics stream files contain device-dependent graphs.

To create a GSF under the MVS system, include the following statements in your program:

```
GOPTIONS DEV=device GSFNAME=GOUT GSFLLEN=80 GSFMODE=REPLACE;

FILENAME GOUT 'aaaaiii.dsn' UNIT=FILE DISP=(MOD,CATLG)
             RECFM=VB SPACE=(TRK,(a,b));
```

where *device* is the SAS/GRAPH device driver name and *a,b* is the number of tracks to allocate. For more information on the FILENAME statement see “*Using SAS at NIH: Batch Mode.*”

It is very important to transfer the graphics stream file as a **binary** file.

The following example can be used on the MVS system to create a graphics stream file to be sent to an HP Laser Jet III that is connected to a PC.

```
//iii JOB (aaaa,999,A),lastname
// JCLLIB ORDER=ZABCRUN.PROCLIB
// EXEC SAS
//SYSIN DD *

FILENAME GOUT 'aaaaiii.TRAFFIC.GSF' UNIT=FILE RECFM=VB
             DISP=(MOD,CATLG) SPACE=(TRK,(10,1),RLSE);
GOPTIONS DEV=HPLJS3 GSFNAME=GOUT GSFLLEN=80 GSFMODE=REPLACE;

DATA TEST;
INPUT YEAR TRAFFIC @@;
CARDS;
90 120 91 140 93 180
;
PATTERN1 V=S C=HWHPJ002;
PATTERN2 V=S C=HWHPJ004;
PATTERN3 V=S C=HWHPJ006;

PROC GCHART;
VBAR YEAR / DISCRETE SUMVAR=TRAFFIC PATTERNID=MIDPOINT;
```


After the file TRAFFIC.GSF is downloaded to the PC as a **binary** file, it can be sent to the HP printer with the following DOS command:

```
copy traffic.gsf lpt1: /b
```

where LPT1 is the port to which the printer is connected to.